

The reference concatenation also applies to whole rows and whole columns. For example `SUM(A:B~D:D)` is the sum of all cells in columns A and B and the column D.

### Reference intersection operator

The intersection operator is written as an exclamation mark. An expression using the intersection operator has the following syntax:

```
reference left ! reference right
```

If the references refer to single ranges, the result is a reference to a single range, containing all cells, which are both in the left reference and in the right reference.

If the references are reference lists, then each list item from the left is intersected with each one from the right and these results are concatenated to a reference list. The order is to first intersect the first item from the left with all items from the right, then intersect the second item from the left with all items from the right, and so on.

### Examples

```
A2:B4 ! B3:D6
```

This results in a reference to the range B3:B4, because these cells are inside A2:B4 and inside B3:D6. This is illustrated in Figure 221, in which the cells in the range A2:B4 have orange backgrounds and the cells in the range B3:D6 have thick black borders. The cells that have both an orange background and a thick black border (B3:B4) form the intersection of the two ranges.

|   | A | B | C | D |
|---|---|---|---|---|
| 1 |   |   |   |   |
| 2 |   |   |   |   |
| 3 |   |   |   |   |
| 4 |   |   |   |   |
| 5 |   |   |   |   |
| 6 |   |   |   |   |
| 7 |   |   |   |   |

Figure 221: Simple example of reference intersection operator

```
(A2:B4~B1:C2) ! (B2:C6~C1:D3)
```

First the intersections `A2:B4!B2:C6`, `A2:B4!C1:D3`, `B1:C2!B2:C6`, and `B1:C2!C1:D3` are calculated. This results in B2:B4, empty, B2:C2, and C1:C2. Then these results are concatenated, dropping empty parts. So the final result is the reference list `B2:B4 ~ B2:C2 ~ C1:C2`.

```
A:B ! 10:10
```

Calculates the intersection of columns A and B with line 10, thus selecting A10 and B10.

You can use the intersection operator to refer a cell in a cross tabulation in an understandable way. If you have columns labeled 'Temperature' and 'Precipitation' and the rows labeled 'January', 'February', 'March', and so on, then the following expression

```
'February' ! 'Temperature'
```

will reference the cell containing the temperature in February.

The intersection operator (!) has a higher precedence than the concatenation operator (~), but do not rely on precedence.



### Tip

Always put in parentheses the part that is to be calculated first.